// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/SekalaAuctionHouse.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.13;

import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/access/Ownable.sol";

interface IERC721 {
    function safeTransferFrom(address from, address to, uint tokenId) external;

    function transferFrom(address, address, uint) external;
}

contract SekalaAuctionHouse is Ownable, ReentrancyGuard {
    error AuctionNotInitialized();
    error AuctionNotLive();
    error AuctionLive();
    error ReservePriceNotMet();
    error WithdrawFailed();
    error NotEOA();
    error IncrementalPriceNotMet();
    error BuyPriceNotMet();
    error CanvasSold();

    event AuctionCreatedUpdated(
        address indexed seller,
        address indexed nftContract,
        uint256 indexed tokenId,
        uint256 startTime,
        uint256 endTime,
        uint256 timeBuffer,
        uint256 reservePrice,
        uint256 minBidIncrementPercentage,
        uint256 buyAmount
    );
    event AuctionPriceUpdated(uint256 indexed auctionId, uint256 reservePrice);
    event Bid(uint256 indexed tokenId, address indexed sender, uint amount);
    event BidIncreased(
        uint256 indexed tokenId,
        address indexed sender,
        uint256 amount
    );
    event Buy(uint256 indexed tokenId, address indexed sender, uint256 amount);
    event AuctionExtended();
    event End(uint256 indexed tokenId, address indexed sender, uint256 amount);

    struct Auction {
        address seller;
        address nftContract;
        uint256 tokenId;
        uint256 startTime;
        uint256 endTime;
        uint256 timeBuffer;
        uint256 reservePrice;
        uint256 minBidIncrementPercentage;
        address bidder;
        uint256 amount;
        uint256 buyAmount;
        bool isSold;
    }

    mapping(uint256 => Auction) public tokenIdToAuction;

    constructor() {}

    modifier onlyEOA() {
        if (tx.origin != msg.sender) {
            revert NotEOA();
        }
        _;
    }

    function createAuctions(Auction[] calldata auctions) external onlyOwner {
        for (uint256 i; i < auctions.length; ) {
            Auction storage auction = tokenIdToAuction[auctions[i].tokenId];

            auction.seller = msg.sender;
            auction.nftContract = auctions[i].nftContract;
            auction.tokenId = auctions[i].tokenId;
            auction.startTime = auctions[i].startTime;
            auction.endTime = auctions[i].endTime;
            auction.timeBuffer = auctions[i].timeBuffer;
            auction.reservePrice = auctions[i].reservePrice;
            auction.minBidIncrementPercentage = auctions[i]
                .minBidIncrementPercentage;
            auction.amount = auctions[i].amount;
            auction.buyAmount = auctions[i].buyAmount;

            emit AuctionCreatedUpdated({
                seller: msg.sender,
                nftContract: auctions[i].nftContract,
                tokenId: auctions[i].tokenId,
                startTime: auctions[i].startTime,
                endTime: auctions[i].endTime,
                timeBuffer: auctions[i].timeBuffer,
                reservePrice: auctions[i].reservePrice,
                minBidIncrementPercentage: auctions[i]
                    .minBidIncrementPercentage,
                buyAmount: auctions[i].buyAmount
            });

            unchecked {
                ++i;
            }
        }
    }

    function updateAuction(
        uint256 tokenId,
        address seller,
        address nftContract,
        uint256 startTime,
        uint256 endTime,
        uint256 timeBuffer,
        uint256 reservePrice,
        uint256 minBidIncrementPercentage,
        uint256 buyAmount
    ) external onlyOwner {
        Auction storage auction = tokenIdToAuction[tokenId];

        auction.seller = seller;
        auction.nftContract = nftContract;
        auction.startTime = startTime;
        auction.endTime = endTime;
        auction.timeBuffer = timeBuffer;
        auction.reservePrice = reservePrice;
        auction.minBidIncrementPercentage = minBidIncrementPercentage;
        auction.buyAmount = buyAmount;

        emit AuctionCreatedUpdated({
            seller: seller,
            nftContract: nftContract,
            tokenId: tokenId,
            startTime: startTime,
            endTime: endTime,
            timeBuffer: timeBuffer,
            reservePrice: reservePrice,
            minBidIncrementPercentage: minBidIncrementPercentage,
            buyAmount: buyAmount
        });
    }

    function bid(uint tokenId) external payable nonReentrant onlyEOA {
        Auction storage auction = tokenIdToAuction[tokenId];

        if (auction.isSold) {
            revert CanvasSold();
        }
        if (auction.startTime == 0 || auction.endTime == 0) {
            revert AuctionNotInitialized();
        }
        if (
            block.timestamp < auction.startTime ||
            block.timestamp > auction.endTime
        ) {
            revert AuctionNotLive();
        }

        if (auction.bidder == msg.sender) {
            // Case when the user already has an active bid
            if (
                msg.value <
                (auction.amount * auction.minBidIncrementPercentage) / 100 ||
                msg.value == 0
            ) {
                revert IncrementalPriceNotMet();
            }

            uint oldValue = auction.amount;
            unchecked {
                auction.amount = oldValue + uint(msg.value);
            }

            emit BidIncreased(tokenId, msg.sender, auction.amount);
        } else {
            if (msg.value < auction.reservePrice || msg.value == 0) {
                revert ReservePriceNotMet();
            }

            // check if there was a previous bid
            if (auction.bidder != address(0)) {
                if (
                    msg.value <
                    auction.amount +
                        (auction.amount * auction.minBidIncrementPercentage) /
                        100
                ) {
                    revert IncrementalPriceNotMet();
                }
            }

            // Refund latest bidder
            if (auction.bidder != address(0)) {
                _transferETH(auction.bidder, auction.amount);
            }

            auction.bidder = msg.sender;
            auction.amount = msg.value;

            emit Bid(tokenId, msg.sender, msg.value);
        }

        // Extend the auction if the bid was received within `timeBuffer` of the auction end time
        if (auction.endTime - block.timestamp < auction.timeBuffer) {
            unchecked {
                auction.endTime = block.timestamp + auction.timeBuffer;
            }
            emit AuctionExtended();
        }
    }

    function buy(uint256 tokenId) external payable nonReentrant onlyEOA {
        Auction storage auction = tokenIdToAuction[tokenId];

        if (auction.isSold) {
            revert CanvasSold();
        }
        if (auction.startTime == 0 || auction.endTime == 0) {
            revert AuctionNotInitialized();
        }
        if (
            block.timestamp < auction.startTime ||
            block.timestamp > auction.endTime
        ) {
            revert AuctionNotLive();
        }

        if (auction.bidder == msg.sender) {
            // Case when the user already has an active bid
            if (
                msg.value < auction.buyAmount - auction.amount || msg.value == 0
            ) {
                revert BuyPriceNotMet();
            }
        } else {
            if (msg.value < auction.buyAmount || msg.value == 0) {
                revert BuyPriceNotMet();
            }

            // Refund latest bidder
            if (auction.bidder != address(0)) {
                _transferETH(auction.bidder, auction.amount);
            }
        }

        emit Buy(tokenId, msg.sender, auction.buyAmount);

        auction.isSold = true;

        IERC721(auction.nftContract).safeTransferFrom(
            auction.seller,
            msg.sender,
            auction.tokenId
        );

        emit End(tokenId, msg.sender, auction.buyAmount);
    }

    function end(uint256[] calldata _tokenIds) external nonReentrant onlyOwner {
        for (uint256 i; i < _tokenIds.length; ) {
            Auction storage auction = tokenIdToAuction[_tokenIds[i]];

            if (auction.isSold) {
                revert CanvasSold();
            }
            if (auction.startTime == 0 || auction.endTime == 0) {
                revert AuctionNotInitialized();
            }
            if (
                block.timestamp > auction.startTime &&
                block.timestamp < auction.endTime
            ) {
                revert AuctionLive();
            }

            auction.isSold = true;

            if (auction.bidder != address(0)) {
                IERC721(auction.nftContract).safeTransferFrom(
                    auction.seller,
                    auction.bidder,
                    auction.tokenId
                );
            }

            emit End(_tokenIds[i], auction.bidder, auction.amount);

            unchecked {
                ++i;
            }
        }
    }

    function _transferETH(address to, uint256 value) internal returns (bool) {
        (bool success, ) = to.call{value: value, gas: 30000}(new bytes(0));
        return success;
    }

    function withdraw() external onlyOwner {
        bool success = _transferETH(msg.sender, address(this).balance);
        if (!success) {
            revert WithdrawFailed();
        }
    }
}